

## Handout Six

November 3, 2011

# 1 Fortran 90 Modules

So far you have only been introduced to one type of ‘program unit’, that is, the main program unit. At first, in the main program unit, you were shown how to write simple executable code and then later how to use a ‘CONTAINS’ statement to include your own functions and subroutines. There is another form of program unit in Fortran 90 and it is called a ‘MODULE’. A module has a range of applications in Fortran 90 and it is good practice to use them. A module can be seen as a related collection of entities that can be made available to one or more other program units. In this case an entity could be data, functions or subroutines. So, for example, you could have a module that contains all the ‘functions’ and ‘subroutines’ you have written that relate to matrices. This matrix module could then be incorporated into any other program you write that makes use of matrices, this provides a neat and structured way of including code you have already written into new code.

## 1.1 The structure of a module

Like a program unit a module has a ‘typical’ structure. This structure is similar to the main program unit.

---

```
MODULE <module name>

  <USE [other modules]>

  IMPLICIT NONE

  <Specification Section>

CONTAINS

  <module procedure one>
  <module procedure two>
  :
  :
  <module procedure n >

END MODULE <module name>
```

---

- A module has its own name like any other program unit.
- A module can also make use of any entities in other modules, so you could have a ‘USE’ statement between the module header and the ‘IMPLICIT NONE’ statement.
- A module has its own ‘IMPLICIT NONE’ statement.
- In the specification part of a module you can declare any data that you want to be ‘globally’ visible to all the procedures defined later in the module.

- A module can contain functions and subroutines just like a main program unit by placing them between a ‘CONTAINS’ statement and the ‘END MODULE’ statement.
- Modules are compiled separately from the main program unit so you should put them in a separate file. You can group more than one module in a single file but in most cases it is sensible to use a separate file for each module. So, for example, you could call the matrix module ‘matrix\_mod.f90’.

## 1.2 Making ‘USE’ of a module

The last section explained what a fortran ‘MODULE’ is, explained its structure and how to write one in fortran 90 code. In this section you will learn how to ‘USE’ a fortran module in your programs so you can gain access to its ‘SUBROUTINES’, ‘FUNCTIONS’ and data.

Any declared entities in a module that are intended to be used externally can be accessed by Fortran 90’s ‘USE’ statement. The ‘USE’ statement can be included in any program unit, function or subroutine of a Fortran 90 code. Its syntax can be described in **three** forms.

### 1.2.1 The simple ‘USE’ statement

USE <module\_name>

This makes accessible all the available entities of the module, so every procedure or item of data declared in the module would be made available. So in an example code this would look like,

---

```
PROGRAM modtest

  !*** Get access to "fact" and "bisect" in module
  USE moduletest

  IMPLICIT NONE

  REAL :: fact_local
  LOGICAL :: bisect_local=.TRUE.
  INTEGER :: a=5
  REAL :: b=0,c=1

  fact_local=fact(a)
  IF (bisect_local) CALL bisect(b,c)

END PROGRAM modtest
```

---

Note that the ‘USE’ statement is included before the ‘IMPLICIT NONE’ statement.

### 1.2.2 The ‘USE’ Statement with the rename ‘=>’ option

It is possible to rename a data type or procedure in a ‘USE’ statement. The reason being that in some cases the ‘USE’ statement, as used above, could cause problems. If a local data item in the program unit using the module is declared with the same name as an entity declared in the module then this would cause a name conflict. A conflict could also exist if a program unit uses more than one module and there is a name conflict between those modules.

For example if we had a module that ‘CONTAINS’ a procedure called ‘fact’ and a program unit that uses the module has a local data type declared with the name ‘fact’, then ‘USE’ of this module as in the

example above will result in a conflict! To get round this problem Fortran 90 uses a re-name symbol ‘=>’ as in;

```
USE <mod_name>, <local_name> => <mod_name>[, <local_name2 > => <mod_name2>]
```

The small example program below uses the module ‘`moduletest`’ and renames the function ‘`fact`’ in the module to be called ‘`factorial`’ locally. It also renames the subroutine ‘`bisect`’ in the module to be called ‘`bisection`’ locally. It does this as both ‘`fact`’ and ‘`bisect`’ already exist as local variables in the main program unit and hence avoids any name conflicts!

---

```
PROGRAM modtest

  USE moduletest, factorial => fact, bisection => bisect

  IMPLICIT NONE

  REAL :: fact
  LOGICAL :: bisect=.TRUE.
  INTEGER :: a=5
  REAL :: b=0,c=1

  fact=factorial(a)
  IF (bisect) CALL bisection(b,c)

END PROGRAM modtest
```

---

### 1.2.3 The ‘USE, ONLY :’ statement

It is clear that it would be advantageous if there was a way to ‘select’ which entities we would like to ‘USE’ from a module. Not surprisingly there is an easy way of doing this and it involves including the ‘ONLY’ attribute inside the ‘USE’ statement.

```
USE <module_name>, ONLY : <name1>[,<name2>,.....]
```

So for the above example we could have.

```
USE moduletest, ONLY : fact, bisect
```

Note that the renaming ‘=>’ can also be used in the comma separated list, for example

```
USE moduletest, ONLY : factorial => fact, bisection => bisect
```

This means we make available from the module ‘`moduletest`’ only the entities `fact` and `bisect` and, in the process, rename them locally `factorial` and `bisection`.

## 1.3 Modules used to hold common data

Modules can be used to pass data between program entities. This is done by declaring the commonly used data in the specification section of the module. It is then simply included in the desired procedure

or program unit using the ‘USE’ statement.

---

```
MODULE comdata

  IMPLICIT NONE

  SAVE
  REAL :: pi=3.14159
  INTEGER :: married=10, age=45

END MODULE comdata

! *****

PROGRAM example

  USE comdata, ONLY : age, married, pi

  IMPLICIT NONE

  PRINT*,age,married,pi

END PROGRAM example
```

---

This method of common data transfer between different sections of code should be not be liberally used instead of argument lists. However, it is often the case that many argument lists in the same code can get very long and complicated and full of the same common data elements. It is in cases like this that common data can be passed to the different sections through modules in this way.

Note that following the ‘IMPLICIT NONE’ statement there is a ‘SAVE’ statement. This is recommended in Fortran 90 to ensure that the data values are preserved between various program units making use of the module. You should always include it in a module when you declare data.

#### 1.4 Some reasons for using Fortran 90 modules

We have just looked at how we can use modules to provide useful entities such as functions, subroutines and also common data types to other program units. Aside from holding some useful entity modules have other uses.

- Modules help keep the main program file at a tidy manageable size.
- Modules allow for better ‘data protection’ by limiting the scope of main program variables.

## 1.5 An Example Code

The first section of the code is the 'module'. This module section goes in a file of it's own and should be named appropriately, 'sin\_module.f90' or 'sin\_mod.f90' for example.

---

```
MODULE series_routines  !***** example module *****

  IMPLICIT NONE

CONTAINS

  FUNCTION expand_sine(x) !*** Finds series expansion

    REAL, DIMENSION(:) :: x
    REAL, DIMENSION(SIZE(x)) :: expand_sine

    expand_sine=x-x**3/factorial(3)+x**5/factorial(5)-&
      x**7/factorial(7)+x**9/factorial(9)

  END FUNCTION expand_sine

! *****

  FUNCTION factorial(n)
  !*** calculates factorials

    INTEGER, INTENT(IN) :: n
    REAL :: factorial, a
    INTEGER :: i

    a=1.0
    DO i = 1,n
      a = a*i
    END DO

    factorial=a

  END FUNCTION factorial

END MODULE series_routines
```

---

The next section of code is the main program unit and this also is best placed in a file of its own. An appropriate name could be something like 'sin\_main.f90'. Of course we still have to compile our source code and then link the object code into an executable file. Note that in earlier codes all the source code was written into a single file and the compiler created an executable file in a single pass. With more than one file, the compiler will need more passes to create the executable. Firstly for each source code file we will need to compile and create an object file. Then all the object files will be linked together to create the executable file. To make this process easier we will be making use of 'makefiles'. A 'makefile' is a set of instructions to the linux 'make' command. The purpose of the 'make' utility is to determine automatically which pieces of a large program need to be compiled or recompiled, and issue the correct commands to do so. The usage of the

'make' command will be explained in more detail shortly.

---

```
PROGRAM sin_main

USE series_routines, ONLY : expand_sine

IMPLICIT NONE

REAL, PARAMETER :: pi = 3.14159265359
REAL, DIMENSION(20) :: x, series
INTEGER :: i

DO i=1,20  !*** Define the range of x values to use
    x(i)=2.0*pi*(i-1)/ 19
END DO

series=expand_sine(x) !**** sin(x) up to five terms

!**** Prints the results to the screen
PRINT '(2e12.4)', (x(i), series(i), i = 1, 20)

END PROGRAM sine_main
```

---

Some points to note about the code.

- The function 'expand\_sine' uses the expression

$$\text{expand\_sine} = x - x^3/\text{factorial}(3) + x^5/\text{factorial}(5) - x^7/\text{factorial}(7) + x^9/\text{factorial}(9)$$

This is an example of array arithmetic, there is nothing complicated here. Arrays of the same **shape** can be added together, subtracted from each other and raised to a power just like scalar numeric data types. When array arithmetic takes place like this each element in each array operand operates on the corresponding element in the other array operand. This is often referred to as an array element by element operation, for example.

$$(1,2,4,3) + (2,5,2,1) = (3,7,6,4)$$
$$(2,1) * (4,2) = (8,2)$$

If for example 'mat1, mat2 and mat3' were matrices of the same shape then the following would be legitimate operations in fortran 90.

```
mat1=mat2+mat3
mat1=mat2-mat3
mat1=mat2*mat3
```

- The expression 'mat2\*mat3' is NOT the same as traditional matrix multiplication i.e. as done by the intrinsic function 'MATMUL'. Instead each element in 'mat2' is multiplied by the corresponding element in 'mat3' and the result stored in the corresponding element of 'mat1'. If an array is operated on by a scalar then the scalar operates on all the elements in that array eg.

$$(1,2,3) * 3 = (3,6,9)$$

- In the main program after the call to `'expand.sine'` we use a `'PRINT'` statement to write the results to the screen. This statement contains an example of an implied `'DO'` loop. They fit on a single line and are convenient in input/output (io) statements and also for initialising arrays. For example the two following statements produce identical results for `'vec2'` and `'vec3'`.

```
'INTEGER :: i'
'INTEGER, DIMENSION(5) :: vec2=(/ (i,i=1,5) /)'
```

This statements declare the arrays and also initialise the elements to be `'1,2,3,4,5'`.

The syntax of an implied DO is the following:

```
'( item-1, item-2, ..., item-n, DO-var = initial, final, step )'
```

It starts with a `'(,'` followed by a set of items, separated by commas, followed by a `'DO'` variable, an equal sign, an initial value, a final value, and a step-size, end ends with a `').'`. Like a typical `'DO'` loop, if the step size is 1, it can be eliminated.

Depending on the place where an implied `'DO'` is used, the items can be variables, including array elements, or expressions.

The meaning of an implied `'DO'` is simple: For each possible value of the `'DO'` variable, all items (i.e., item-1, item-2, ..., item-n) are listed once and adjacent items are separated by commas. The following small code will help you follow how the syntax of the implied `'DO'` loop work.

```
PROGRAM implied
!**
!** Example of an implied DO loop

    IMPLICIT NONE

    INTEGER :: i
    REAL, DIMENSION(8) :: aa=(/(i,2*i,i=1,4)/)

    PRINT' (f4.1)',(aa(i),i=1,8)
```

```
END PROGRAM implied
```

```
!* The output to the screen is as below
```

```
1.0
2.0
2.0
4.0
3.0
6.0
4.0
8.0
```

The above code `'sine_expansion'` demonstrates how you can use modules to build a code in a modular fashion. You can copy this code, and any files needed to compile it, to a directory in your home area by typing

```
cp ~/info/examples/sinmod/* .
```

in the directory you want to copy the files to. Do not forget the dot `'.'` at the end to indicate you

want the file to be copied to the directory you are currently in. To compile and run the code you will have to read the next section that introduces the 'Makefile' at a very basic level.

### 1.5.1 Compiling more than one file : 'Makefile'

If you copied the files as directed in the above section you will now have in your directory a file called 'Makefile'. This file allows you to compile the two source code files and link together the object files (created by the compilation) into a single executable by just using a single command 'make'. Try it, just type 'make' and hit [Return]. You can look at the file 'Makefile' in 'nano'. You do not need to understand any of the main part you just have to know how to change the first section that contains the lines 'main=sin\_main' and 'mod1=sin\_module'.

---

```
#####
# REPLACE the main_prg_name below with the name of
# your main program without the ".f90" extension
# REPLACE the module_name below with your module
# name without the ".f90" extension

# PROGRAM NAME (no .f90)
main=sin_main
# MODULE NAME (no .f90)
mod1=sin_module

# compiler options
opt1=-check all -check noarg_temp_created
opt2=-check nouninit -traceback
#####
cmplr=f90 $(opt1) $(opt2)
objects1 = $(mod1).o $(main).o

$(main) : $(objects1)
    $(cmplr) -o $(main) $(objects1)
$(mod1).o : $(mod1).f90
    $(cmplr) -c $(mod1).f90
$(main).o : $(main).f90 $(mod1).f90
    $(cmplr) -c $(main).f90

clean :
    rm -f *.mod *.pcl *.pc *.o $(main) *.inc *.vo *.d
```

The 'Makefile' contains 'required' tabs at the beginning of certain lines so please copy 'cp' this file around instead of typing it in yourself!

---

This makefile will compile and link any code that has a main program file and a module. At the moment the 'Makefile' is setup for a main program file called 'sin\_main.f90' and a module file called 'sin\_module.f90'. You will want to compile and link codes that have different names for the main program unit and the 'Makefile'. To do this you must copy the 'Makefile' to the directory containing the main program file and the module it uses. Then change the lines 'main=sin\_main' and 'mod1=sin\_module' to 'main=<newname1>' and 'mod1=<newname2>'. Where 'newname1' and 'newname2' are your main program file name and your module file name without the '.f90' extension.

---

### Class Project :: Part Three :

Create a new directory for your matrix library code and copy your class project fortran file into it, this is so you will still have a copy of the old one. Rearrange your matrix fortran source code file into two 'new' files, one main program file and the other a 'MODULE' file containing all the procedures you have written so far. Call the module 'matrix\_mod.f90' and in addition to the procedures you have already written, add the following procedures to your 'matrix\_mod.f90' file.

- Write a Function called 'transmat' to return the transpose of a matrix. The transpose of a matrix  $\mathbf{A}$  is written  $\mathbf{A}^T$  and is simply  $\mathbf{A}$  with the rows and columns interchanged. So row one in  $\mathbf{A}^T$  is column one in  $\mathbf{A}$  etc.
- Write a Function called 'normtwo' to return the Euclidean norm, ( $\|\mathbf{v}\|_2$ ) of a vector  $\mathbf{v}$ . The Euclidean norm is often referred to as the length of a vector and is the square root of the sum of the squares of all the elements in the vector.

### The Power Method

Let  $\mathbf{A}$  be an  $m \times m$  real matrix with  $m$  real eigenvalues. Moreover, assume that  $\mathbf{A}$  has precisely one eigenvalue ( $\lambda_1$ ) that is 'dominant' (largest in absolute magnitude) with a corresponding eigenvector ( $\mathbf{x}$ ). Then  $\mathbf{x}$  and  $\lambda_1$  can be calculated using the 'power method' for approximating eigenvalues. The 'power method' is an iterative method and calculates a new estimate of the eigenvector  $\mathbf{x}^{(n+1)}$  from the previous estimate  $\mathbf{x}^{(n)}$ .

[Q] How does the 'power method' work?

[A] It works by taking an initial 'guestimate' at the eigenvector and then repeatedly premultiplying this guess with the matrix  $\mathbf{A}$ . The  $m$  eigenvectors of our  $m \times m$  real matrix  $\mathbf{A}$  form a set of linearly independent basis vectors in  $\mathbb{R}^m$   $\{\mathbf{v}_1, \mathbf{v}_2, \dots, \mathbf{v}_m\}$ . The key to understanding this method is by making use of the fact that the arbitrary vector  $\mathbf{x}$  (our initial guess) can be written as a linear combination of these  $m$  eigenvectors. Therefore,

$$\mathbf{x} = c_1\mathbf{v}_1 + c_2\mathbf{v}_2 + \dots + c_m\mathbf{v}_m \quad (1)$$

$$\mathbf{Ax} = c_1\lambda_1\mathbf{v}_1 + c_2\lambda_2\mathbf{v}_2 + \dots + c_m\lambda_m\mathbf{v}_m \quad (2)$$

$$\mathbf{A(Ax)} = \mathbf{A}^2\mathbf{x} = c_1\lambda_1^2\mathbf{v}_1 + c_2\lambda_2^2\mathbf{v}_2 + \dots + c_m\lambda_m^2\mathbf{v}_m \quad (3)$$

$$\mathbf{A}^n\mathbf{x} = c_1\lambda_1^n\mathbf{v}_1 + c_2\lambda_2^n\mathbf{v}_2 + \dots + c_m\lambda_m^n\mathbf{v}_m \quad (4)$$

The equations 1  $\rightarrow$  4 show how repeatedly premultiplying  $\mathbf{x}$  with  $\mathbf{A}$  builds up the  $m$  terms involving the  $m$  eigenvectors. Now since the first term on the RHS of (4) contains  $\lambda_1$ , the dominant (largest in absolute value) eigenvalue, as  $n$  gets larger this first term will start to dominate the other terms in absolute value. So the RHS will eventually converge to our dominant eigenvector as the first term containing  $\mathbf{v}_1$  is providing the largest contribution to the summation. There is a problem, however, that as  $n$  gets large  $\lambda^n$  gets very large in fact too large for the computer to accurately represent its value. So, to solve this problem, after each time we premultiply by  $\mathbf{A}$  we normalise the result to keep our new estimate of the dominant eigenvector of order one. Remember that dividing an eigenvector by a scalar still leaves the same eigenvector but simply rescaled.

The main iterative body of power method can be written as;

- Take a normalised estimate of the eigenvector  $\mathbf{x}^{(n)}$  and calculate  $\mathbf{y}^{(n)} = \mathbf{A} \cdot \mathbf{x}^{(n)}$ .
- Calculate the new estimate of the dominant eigenvalue  $\lambda_1^{(n)} = \mathbf{y}_k^{(n)} / \mathbf{x}_k^{(n)}$ , where  $k \in \{1, 2, \dots, m\}$
- Calculate the new normalised  $(n+1)^{th}$  estimate of the dominant eigenvector from the  $(n)^{th}$  estimate of  $\mathbf{y}$

$$\mathbf{x}^{(n+1)} = \frac{\mathbf{y}^{(n)}}{\|\mathbf{y}^{(n)}\|_\infty}$$

- Repeat the above three steps until a termination criterion has been satisfied. Note that  $\mathbf{x}^{(n)}$  means the  $n^{\text{th}}$  iterative estimate **not**  $\mathbf{x}$  raised to the power  $n$ .

(a) In your matrix library ‘MODULE’ add the Fortran 90 procedures, described below.

- (i) First write a ‘FUNCTION’ called ‘`infnorm`’ that calculates and returns the ‘infinity’ norm of a vector  $\mathbf{v} \in \mathbb{R}^m$ . The infinity norm,  $\|\mathbf{v}\|_\infty$ , is simply the modulus of the element in  $\mathbf{v}$  with the largest absolute value.

- (ii) Write a ‘FUNCTION’ called ‘`cont`’ that accepts **only** the four dummy arguments

‘`(y,x,tol,max_iters)`’

and returns ‘.FALSE.’ if a given tolerance has been met or a maximum number of iterations has been exceeded else it returns ‘.TRUE.’. Use for the tolerance condition,

$$\|\mathbf{x}^{(n+1)} - \mathbf{x}^{(n)}\|_\infty < \text{tol}$$

use `tol = 0.001`.

HINT : Declaring a variable inside a procedure with the attribute ‘SAVE’, means that the value of that variable will be retained between calls to that procedure.

```
INTEGER, SAVE :: count=1
```

declares ‘`count`’ to be initialised with the value ‘1’ only on the first call to the procedure it is declared in, from then on it retains its value between calls.

- (iii) Write a ‘SUBROUTINE’ called ‘`power`’ that has as its argument list;

‘`(mat,x,tol,eigv,conv)`’

Where ‘`mat`’ is the matrix for which the dominant eigenvalue is to be found, ‘`x`’ is to input the initial eigenvector estimate and return the final estimate, ‘`tol`’ is the measure of convergence, ‘`eigv`’ is to hold the returned eigenvalue and ‘`conv`’ is of type ‘LOGICAL’ and returns ‘.TRUE.’ only if the method converged.

- (iv) You will need a function ‘`mulmatvec(mat,vec)`’ that pre-multiplies a vector by a matrix’

Use you code to calculate the dominant eigenvalue and corresponding eigenvector of the matrix;

$$\begin{pmatrix} 1 & 5 & 3 \\ 6 & 3 & 5 \\ 2 & 8 & 5 \end{pmatrix}$$

You should have a dominant eigenvalue of 13.063. with a corresponding eigenvector of

$$\begin{pmatrix} 0.6039 \\ 0.8569 \\ 1.0000 \end{pmatrix}$$